
1. Introduction

This document provides basic information on using the Nimbus-5 Scanning Chopper Radiometer 

(SCR) Level-1 Radiance Data product.

1.1 Data Product Description

The Nimbus-5 Selective Chopper Radiometer (SCR) Level-1 Radiance Data product contains 

calibrated, earth-located radiances that were prepared by the experimenter's office on binary 

magnetic tapes. The radiances, measured by 16 channels have a ground resolution of 25 km, are 

"declouded" (interpolated and smoothed across regions of cloud). They are grouped into major 

frames along with orbit, altitude, latitude, longitude, and some ancillary data.

Each file contains one day of data (~14 orbits per day). Spatial coverage is near global between 

latitude -80 and +80 degrees. The data are available from 1972/12/13 (day of year 348) through 

1974/12/26 (day of year 360).

This product was previously available from the NASA National Space Science Data Center (NSSDC) 

under the name Selective Chopper Radiometer Radiance Data with the identifier ESAD-00250 (old id

72-097A-02A).

1.1.1 The Selective Chopper Radiometer

The Nimbus 5 Selective Chopper Radiometer (SCR) was designed to observe the global temperature 

structure of the atmosphere up to 50 km in altitude, make supporting observations of water vapor 

distribution, and determine the density of ice particles in cirrus clouds. To accomplish these 

objectives, the SCR measured emitted radiation in 16 spectral intervals separated into the following 

four groups: (1) four CO2 channels between 13.8 and 14.8 microns, (2) four channels at 15.0 

microns, (3) an IR window channel at 11.1 microns, a water vapor channel at 18.6 microns, two 

channels at 49.5 and 133.3 microns, and (4) four channels at 2.08, 2.59, 2.65,and 3.5 microns. From 

an average satellite altitude of 1100 km, the radiometer viewed a 48-km circle on the earth's surface

with a ground resolution of about 25 km. The instrument operated successfully for about two years. 

A similar experiment was flown on the previous Nimbus 4 satellite.

The principal investigator for the SCR experiment was Prof. John T. Houghton from Oxford 

University.

1.1.2 Nimbus-5 Overview

The Nimbus 5 satellite was successfully launched on December 11, 1972. The primary experiments
included:  (1)  a  Temperature-Humidity  Infrared  Radiometer  (THIR)  for  measuring  day  and  night
surface and cloud top temperatures, as well as the water vapor content of the upper atmosphere,
(2) an Electrically Scanning Microwave Radiometer (ESMR) for mapping the microwave radiation
from the Earth's surface and atmosphere, (3) an Infrared Temperature Profile Radiometer (ITPR) for
obtaining vertical profiles of temperature and moisture, (4) the Nimbus-E Microwave Spectrometer
(NEMS) for determining tropospheric temperature profiles, atmospheric water vapor abundances,
and cloud liquid water contents, (5) a Selective Chopper Radiometer (SCR) for observing the global
temperature  structure  of  the  atmosphere,  and (6)  a  Surface  Composition Mapping  Radiometer
(SCMR) for measuring the differences in the thermal emission characteristics of the Earth's surface.

The orbit of the satellite can be characterized by the following:

 circular orbit at 1100 km

 inclination of 99.9 degrees

 period of an orbit is about 107.2 minutes

 orbits cross the equator at 26 degrees of longitude separation

 sun-synchronous

1.2 Algorithm Background

The Nimbus-5 SCR data were generated from the spacecraft telemetry, attitude and orbital data.
The data were originally processed on IBM 360 computers using a 32-bit architecture, and copied to
1600 bpi 9-track tapes for archival.  Further information on the SCR instrument and data processing
can be found in the Nimbus-5 Users' Guide Section 6.

1.3 Data Disclaimer

The data should be used with care and one should first read the Nimbus-5 User’s Guide, section 6 

describing the SCR experiment. Users should cite this data product in their research.

2. Data Organization

The Nimbus-5 Selective Chopper Radiometer Level-1 Radiance Data spans the time period from 

December 13, 1972 to December 26, 1974. Each file contains one day (~14 orbits) of data.

2.1 File Naming Convention

The data product files are named according to the following convention:

<Platform>-<Instrument>_<Product>_<DateTime>_<TapeNumber>.<Suffix>

where:
o Platform = name of the platform or satellite (always Nimbus5)
o) Instrument = name of the instrument and product (always SCR)
o) Product = process level (L1)
o Date = Data start date and time in UTC in format <YYYY>m<MMDD>t<hhmmss> where

1. YYYY = 4 digit year (1970 – 1973)
2. MM = 2 digit month (01-12)
3. DD = 2 digit day of month (01-31)
4. hh = 2 digit hour (00-23)
5. mm = 2 digin minute (00-23)
6. ss = 2 digit second (00-59)

o TapeNumber = 3 digit number of tape (preceeded by ‘DR’ - primary or ‘DS’ - backup)
o Suffix = the file format (always TAP, indicating tape binary data)

File name example: Nimbus5-SCR_L1_1972m1213t023824_DR256.TAP

2.2 File Format and Structure

The data are stored as they were originally written in IBM binary (big-endian) record oriented 
structured files. The files were written on the original 1600 bpi 9-track tapes using a blocked 
FORTRAN format. Note, the first record contains just the first FORTRAN record size header word and
contains the Orbit Header. Subsequent records contain both the FORTAN record leading and trailing 
size header words. A file typically contains a full day of data or 14 orbits.

Each day file contains a day header record/file (with calibration information and size 190 bytes), 
followed by up to 16 orbit files, and ending with an end of day record/file. An orbit file starts with an
Orbit Header Record (size 330 bytes), followed by a set of Data Records (up to size 3714 bytes), and 
ending with a End of Orbit Record (size 18 bytes). Each Data Record consists of up to 10 logical 
records or major frames with radiance measurements. For the contents and layout of the 
documentation, see section 3.1

The documentation says the Orbit Header contains 44 housekeeping functions, however the data 

files indicate this number is actually 48. We do not have documentation explaining this discrepancy 

or what the other 4 functions represent.

There are 620 unique daily data files in this collection, 506 are from the DR primary tapes (9 total), 

the remaining 114 are from the DS backup tapes (13 total). The file for day 1974/02/01 is mostly 

duplicated, but has some records only in the primary copy and other records only in the backup tape

copy.

2.3 Key Science Data Fields

The primary science data fields in this data product are the SCR calibrated radiances in units of 

mW/m2/sr/cm-1 for each channel.

Figure 1: Typical data coverage of a Nimbus 5 SCR Level 1 daily data file.

3. Data Contents

The granularity of this data product is one day (with approx. 14 orbits).

3.1 Data Records

The Nimbus-5 User’s Guide does not describe the layout of the file format. Refer instead to the data 

provider’s documentation “Clarendon Laboratory Atmospheric Physics, The Selective Chopper 

Radiometer on Nimbus V Archived Data”.

The original tape files each included a tape summary file consisting of a summary header, and 

summary day records. These were then followed by a set of up to thirty day header record files, 

which would include a set of up to 16 orbit files and an end of day record file. As part of the 

recovery, the GES DISC has extracted and archived the day and orbit files as a single daily file. The 

tape summary files are not included. The original data were written using 6-bit bytes. During tape 

data recovery these were saved as 8-bit bytes (bits 6 and 7 are thus unused), a word is two 6-bit 

bytes. Characters are in the IBM 48 character BCD encoding. The tables below summarize the file 

record types. Each record type below starts and ends with a four byte word giving the record size in 

bytes.

Table 3-1-1: Day Header Record (190 bytes)

Word Field Name Units Type Comments

0, 1 Initial Two Words 2(I*2) Octal: 7106,7106 

2 Record Length (number of words) I*2

3 Record Number I*2

4 Record Identifier I*2 Octal: 5203

5 Day of Year - I*2

6 Year - I*2

7, 8 Number of Major Frames in Day I*4

9 Number of CSE’s on Day’s Transmission I*2

10 Number of CSE’s Read from Daily Tape I*2

11 Number of Calibration Sequences in Day I*2

12 Number of Orbits in Day I*2

13 – 92 Calibration Data Used for Processing of Day’s Observations 80(I*2)
See the SCR 
Archived Data 
document

93 End-of-Record I*2

94 Checksum I*2

This block is followed by a 4 byte end of file (EOF) marker as this was a separate file on the original tape.

Table 3-1-2: Orbit Header Record (330 bytes)

Word Field Name Units Type Comments

0, 1 Initial Two Words 2(I*2) Octal: 7106,7106 

2 Record Length (number of words) I*2

3 Record Number I*2

4 Record Identifier I*2 Octal: 5204

5,6 Orbit Number - 2(I*2)

7 HDRSS I*2

8 Number of First Major Frames in Orbit I*2

9 Day Number of First Major Frame - I*2

10, 11 Time (GMT) of First Major Frame sec 2(I*2)

12 Day Number of Last Major Frame - I*2

13, 14 Time (GMT) of Last Major Frame sec 2(I*2)

15 Number of CSE’s on Orbit’s Transmission I*2

16 Number of CSE’s Read from Daily Tape I*2

17 Number of Calibration Sequences in Orbit I*2

18 Number of Housekeeping Functions (N=48) I*2 (see note)

19 - 66 N Maximum Values for this Orbit N(I*2)

67 - 114 N Minimum Values for this Orbit N(I*2)

115 -
162

N Mean Values for this Orbit N(I*2)

163 End-of-Record I*2

164 Checksum I*2

Note: The SCR Archived Data document says the value is 44. The data files have a value of 48. Its 

unclear at this time what the other 4 housekeeping functions are.

Table 3-1-4: Data Record (max 3714 bytes; words 8-193 repeats with number given by word 5)

Word Field Name Units Type Comments

0, 1 Initial Two Words 2(I*2) Octal: 7106,7106 

2 Record Length (number of words) I*2

3 Record Number I*2

4 Record Identifier I*2 Octal: 4205

5 Number of Major Frames in this Block - I*2 (max 10)

6 Length of Each Major Frame - I*2 186

7 Spare I*2

8 Checksum Word for Major Frame I*2

9, 10 Orbit Number - 2(I*2)

11 Block Number - I*2

12 Frame Number / HDRSS - I*2

13 Day Number - I*2

14, 15 Time (GMT) sec 2(I*2)

16 Latitude (-90,90) degrees I*2 Scaled by 8

17 Longitude(0-360)
degrees I*2 Scaled by 8

18 Altitude km I*2 Scaled by 8

19 ESMR Maximum I*2

20 ESMR Minimum I*2

21 - 25 Major Frame Flags 5(I*2) Bit encoded

26-74 Calibrated Radiances mW/m2/sr/cm 49(I*2)
See provider’s 
documentation

75 - 79 16 sec Ramps 5(I*2)
See provider’s 
documentation

80 -
123

4 sec Ramps 44(I*2)
See provider’s 
documentation

124 -
128

Digital A Housekeeping 5(I*2)
See provider’s 
documentation

129 -
167

Analog Housekeeping 49(I*2)
See provider’s 
documentation

168 FOVC Ramp I*2

169 -
176

Raw ESMR 8(I*2)

177 Pitch I*2

178 Roll I*2

178 Yaw I*2

180 -
182

A2, A3, A4 Declouded 3(I*2)

183 -
185

B1-B2, B2-B3, B4-B3 Smoothed 3(I*2)

186 Sea Surface Temperature / Geography -deg C or ft I*2
Land = height / 
Ocean = 
temperature

187 -
193

B1, B2, B3, B4, B1-B2, B2-B3, B3-B4 Corrected Radiances 7(I*2)

Word2
-2

End-of-Record I*2

Word2
-1

Checksum I*2

Table 3-1-5: End of Orbit Record (18 bytes with last two words null padded)

Word Field Name Units Type Comments

0, 1 Initial Two Words 2(I*2) Octal: 7106,7106 

2 Record Length (number of words) I*2

3 Record Number I*2

4 Record Identifier I*2 Octal: 5206

5 End-of-Record I*2

6 Checksum I*2

This block is followed by a 4 byte end of file (EOF) marker as this was a separate file on the original tape.

Table 3-1-6: End of Day Record (18 bytes with last two words null padded)

Word Field Name Units Type Comments

0, 1 Initial Two Words 2(I*2) Octal: 7106,7106 

2 Record Length (number of words) I*2

3 Record Number I*2

4 Record Identifier I*2 Octal: 5207

5 End-of-Record I*2

6 Checksum I*2

This block is followed by a 4 byte end of file (EOF) marker as this was a separate file on the original tape.

Notes, the End of Record (EOR) word is defined with the octal values:

 5252 last record of file
 5225 file containing one record
 6453 last record on the tape
 4421 all other records

3.2 Metadata

The metadata are contained in a separate XML formatted file having the same name as the data file 

with .xml appended to it.

Table 3-2: Metadata attributes associated with the data file.

Name Description

LongName Long name of the data product.

ShortName Short name of the data product.

VersionID Product or collection version.

GranuleID Granule identifier, i.e. the name of the file.

Format File format of the data file.

CheckSumType Type of checksum used.

CheckSumValue The value of the calculated checksum.

SizeBytesDataGranul

e

Size of the file or granule in bytes.

InsertDateTime Date and time when the granule was inserted into the archive.  The format for date is YYYY-

MM-DD and time is hh-mm-ss.

ProductionDateTime Date and time the file was produced in format YYYY-MM-DDThh:mm:ss.ssssssZ

RangeBeginningDate Begin date when the data was collected in YYYY-MM-DD format.

RangeBeginningTime Begin time of the date when the data was collected in hh-mm-ss format.

RangeEndingDate End date when the data was collected in YYYY-MM-DD format.

RangeEndingTime End time of the date when the data was collected in hh-mm-ss format.

PlatformShortName Short name or acronym of the platform or satellite

InstrumentShortNam

e

Short name or acronym of the instrument 

SensorShortName Short name or acronym of the sensor

WestBounding 

Coordinate

The westernmost longitude of the bounding rectangle(-180.0 to +180.0)

NorthBounding 

Coordinate

The northernmost latitude of the bounding rectangle(-90.0 to +90.0)

EastBounding 

Coordinate

The easternmost longitude of the bounding rectangle(-180.0 to +180.0)

SouthBounding 

Coordinate

The southernmost latitude of the bounding rectangle(-90.0 to +90.0)

Orbit Satellite orbit number.

ElapsedMinTime Duration in minutes of data collected during an orbit.

4. Reading the Data

The data are written in a binary record-oriented format. Using the record format specification in the 

section above, users can write software to read the data files. Please note that the data were 

originally written using a big-endian format, therefore users on little-endian machines will need to 

swap bytes for the words.

A sample FORTRAN program is included in the Appendix section which will read in the data records. 

Additionally a FORTRAN function is included to perform byte swapping.

5. Data Services

5.1 GES DISC Search

The GES DISC provides a keyword, spatial, temporal and advanced (event) searches through its 

unified search and download interface:

https://disc.gsfc.nasa.gov/

5.2 Documentation

The data product landing pages provide information about these data products, as well as links to 
download the data files and relevant documentation:

https://disc.gsfc.nasa.gov/datacollection/SCRN5L1RAD_001.html

5.3 Direct Download

These data products are available for users to download directly using HTTPS:

https://acdisc.gesdisc.eosdis.nasa.gov/data/Nimbus5_SCR_Level1/SCRN5L1RAD.001/

6. More Information

6.1 Contact Information

Name: GES DISC Help Desk

URL: https://disc.gsfc.nasa.gov/

E-mail: gsfc-help-disc@lists.nasa.gov

Phone: 301-614-5224

Fax: 301-614-5228

Address: Goddard Earth Sciences Data and Information Services Center

Attn: Help Desk

Code 610.2

NASA Goddard Space Flight Center

Greenbelt, MD  20771, USA

6.2 References

"The Nimbus-5 User’s Guide - Section 6: The Selective Chopper Radiometer (SCR) Experiment", 

NASA Goddard Space Flight Center, November 1972, Pages 131-140

Clarendon Laboratory Atmospheric Physics, “The Selective Chopper Radiometer on Nimbus V 

Archived Data”, Atmospheric Physics Memorandum No. 77.1, University of Oxford, 1975

Acronyms

BCD: Binary Coded Decimal

EOS: Earth Observing System

ESDIS: Earth Science and Data Information System

GES DISC: Goddard Earth Sciences Data and Information Services Center

GSFC: Goddard Space Flight Center

HDRSS: High Data Rate Storage System

L1: Level-1 Data

NASA: National Aeronautics and Space Administration

QA: Quality Assessment

SCR: Selective Chopper Radiometer

ESMR: Temperature-Humidity Infrared Radiometer

UT: Universal Time

FORTRAN Code
C-----------------------------------------------------------------------
C ^NAME: READ_SCRN5
C   This program will read an entire Nimbus 5 SCR Level-1 data file.
C
C   The SCR data contain an initial orbit header record, followed by a
C   housekeeping record, series of data records. Each of
C   data record contains a data header and a set of six scan line records.
C   The function can optionally display the temperature data on a cylindrical
C   map, and this can then be sent to an output PNG file. The geolocation
C   metadata may also be overlayed on the map.
C
C ^MAJOR VARIABLES:
C     FNAME  - name of input file
C     BLOCK  - buffer for data block typically has three data records
C     BUFF   - buffer for holding temporary 4-byte word
C     WORD   - integer 4-byte word
C     IBLKSZ - size of block in bytes
C     IOS    - I/O status number
C
C ^NOTES:
C     Compile: gfortran -o READ_SCRN5.EXE READ_SCRN5.FOR
C
C ^ORGANIZATION: NASA/GSFC, Code 610.2
C
C ^AUTHOR:  James Johnson
C
C ^ADDRESS: james.johnson@nasa.gov
C
C ^CREATED: June 7, 2018
C-----------------------------------------------------------------------

      CHARACTER          FNAME*1024
      CHARACTER          BLOCK(10000)  ! Buffer = 8064 bytes
      INTEGER*2          IWORD(5000)
      CHARACTER          BUFF(4)       ! Buffer to hold 4-byte word
      INTEGER*4          WORD          ! 4-byte word
      INTEGER*4          IBLKSZ        ! Block size header
      INTEGER*4          IRECSZ        ! Size of logical records
      INTEGER*4          NRECS         ! Number of logical records
      EQUIVALENCE        (BUFF, WORD)

C     Get the name of the input data file to read
      WRITE (0, *), 'Enter the name of the input file:'
      READ (5,'(A)') FNAME
      PRINT '("File = ",A)', FNAME

C     Open the specified input file
      OPEN (UNIT=1, FILE=FNAME, STATUS='OLD', ACCESS='DIRECT',
     &      FORM='UNFORMATTED', RECL=1, ERR=99, IOSTAT=IOS)

C     Initialize N (block number) and IOFF (byte offset in file)
      N=1
      IOFF=0

C     Loop through the file reading all blocks of data
    5 DO
C       Read the first 4-byte word or block size header
        DO I=1,4
          READ (1, REC=IOFF+I, IOSTAT=IOS, ERR=90) BUFF(I)
        END DO
        IBLKSZ = WORD
        IOFF=IOFF+(I-1)

        IF (IBLKSZ .EQ. 0) THEN
C         PRINT '("WARNING: END-OF-TAPE MARK")'
          GOTO 5
        ENDIF

C       Next read the block of data
        DO I=1,IBLKSZ
          READ (1, REC=IOFF+I, IOSTAT=IOS) BLOCK(I)
          IF (IOS .NE. 0) THEN
            PRINT '("ERROR: BLOCK ",I4,X,I4,", IOSTAT: ",I6)', N,I-1,IOS
            IBLKSZ = I-1
            GOTO 20
          ENDIF

C         Data were written as 12-bit words, store in 2-byte integers
          IF (MOD(I+1,2) .EQ. 0) THEN
            IWORD((I+1)/2) = 0          ! Initialize IWORD to zero
            IWORD((I+1)/2) = IWORD((I+1)/2) + ISHFT(ICHAR(BLOCK(I)),6)
          ELSE
            IWORD((I+1)/2) = IWORD((I+1)/2) + ICHAR(BLOCK(I))
          ENDIF

        END DO
        IOFF=IOFF+(I-1)
        N=N+1

C       The record identifier is word 5
        IF      (IWORD(5) .EQ. '5200'O) THEN
          CALL PR5200(IWORD, IBLKSZ, N)
        ELSE IF (IWORD(5) .EQ. '5201'O) THEN
          CALL PR5201(IWORD, IBLKSZ, N)
        ELSE IF (IWORD(5) .EQ. '5202'O) THEN
          CALL PR52EN(IWORD, IBLKSZ, N)
        ELSE IF (IWORD(5) .EQ. '5203'O) THEN
          CALL PR5203(IWORD, IBLKSZ, N)
        ELSE IF (IWORD(5) .EQ. '5204'O) THEN
          CALL PR5204(IWORD, IBLKSZ, N)
        ELSE IF (IWORD(5) .EQ. '5205'O) THEN
          CALL PR5205(IWORD, IBLKSZ, N)
        ELSE IF (IWORD(5) .EQ. '5206'O) THEN
          CALL PR52EN(IWORD, IBLKSZ, N)
        ELSE IF (IWORD(5) .EQ. '5207'O) THEN
          CALL PR52EN(IWORD, IBLKSZ, N)
        ELSE
          PRINT '("Unknown record identifier: ", O4.4)', IWORD(5)
        ENDIF

Page | 21 



C       Finally read the last 4-byte word
C       (should match first block size)
   20   DO I=1,4
          READ (1, REC=IOFF+I, IOSTAT=IOS, ERR=90) BUFF(I)
        END DO
        IF (IBLKSZ .NE. WORD) THEN
          PRINT '("WARNING: IBLKSZ ",I10," != ",I10)', WORD, IBLKSZ
        ENDIF
        IOFF=IOFF+(I-1)

      END DO

C     Close the input file
   90 CLOSE(1)
      GOTO 100

   99 PRINT '("ERROR: OPEN FILE, IOSTAT: ",I6)', IOS

  100 STOP
      END

C-----------------------------------------------------------------------
C     This Subroutine will print Summary Head Record (5200)
C-----------------------------------------------------------------------
      SUBROUTINE PR5200(WRDARR, IBLKSZ, N)

      INTEGER*2          WRDARR(500)   ! Word Array
      CHARACTER          BUFF*4        ! Temporary data buffer

      PRINT '("******************")'

C     First two words are some kind of file marker
      PRINT '("SYNC   =",2(X,O4.4))', WRDARR(1:2)
C     Record length
      PRINT '("RECLEN =",3X,I6)', WRDARR(3)
C     Record number
      PRINT '("RECNUM =",3X,I6)', WRDARR(4)
C     Record identifier
      PRINT '("RECID  =",5X,O4.4)', WRDARR(5)
C     Day number
      PRINT '("NDAYS  =",3X,I6)', WRDARR(6)
C     EOR
      PRINT '("EOR    =",5X,O4.4)', WRDARR(7)
C     Checksum
      PRINT '("CHKSUM =",3X,I6)', WRDARR(8)

      RETURN
      END

C-----------------------------------------------------------------------
C     This Subroutine will print Summary Day Record (5201)
C-----------------------------------------------------------------------
      SUBROUTINE PR5201(WRDARR, IBLKSZ, N)

      INTEGER*2          WRDARR(500)   ! Word Array
      CHARACTER          BUFF*4        ! Temporary data buffer
      INTEGER*4          I4BUFF        ! 4-byte integer buffer
      CHARACTER          HDRSS(2)

      PRINT '("******************")'

C     First two words are some kind of file marker
      PRINT '("SYNC   =",2(X,O4.4))', WRDARR(1:2)
C     Record length
      PRINT '("RECLEN =",3X,I6)', WRDARR(3)
C     Record number
      PRINT '("RECNUM =",3X,I6)', WRDARR(4)
C     Record identifier
      PRINT '("RECID  =",5X,O4.4)', WRDARR(5)
C     Day number
      PRINT '("DAY    =",3X,I6)', WRDARR(6)
C     Year
      PRINT '("YEAR   =",3X,I6)', WRDARR(7)
C     Number of major frames in day
      I4BUFF = ISHFT(WRDARR(8),12) + WRDARR(9)
      PRINT '("NMFDAY =",3X,I6)', I4BUFF
C     Number of Cse's on Day's Transmission
      PRINT '("NCSEDX =",3X,I6)', WRDARR(10)
C     Number of Cse's Read from Daily Tape
      PRINT '("NCSEDT =",3X,I6)', WRDARR(11)
C     Number of Cal Sequences in Day
      PRINT '("NCALSD =",3X,I6)', WRDARR(12)
C     Number of Orbits in Day (max 16)
      PRINT '("NORBIT =",3X,I6)', WRDARR(13)
      DO I=1,WRDARR(13)
        IOFF=(I-1)*13
        PRINT '("------------------")'
C       Orbit number
        I4BUFF = ISHFT(WRDARR(IOFF+14),12) + WRDARR(IOFF+15)
        IF (I4BUFF .LT. 0) THEN
          I4BUFF = 65536 + I4BUFF
        ENDIF
        PRINT '("ORBNUM =",X,I8)', I4BUFF
C       HDRSS 
        J = ISHFT(WRDARR(IOFF+16),-6)
        HDRSS(1) = CHAR(IBCD(J))
        K = IAND(WRDARR(IOFF+16),'003F'Z)
        HDRSS(2) = CHAR(IBCD(K))
        PRINT '("HDRSS  =",5X,2(X,A1))', HDRSS
C       Number of Major Frames in Orbit
        PRINT '("NMFORB =",3X,I6)', WRDARR(IOFF+17)
C       Day of First Major Frame
        PRINT '("DAY1ST =",3X,I6)', WRDARR(IOFF+18)
C       Time (GMT) of First Major Frame
        I4BUFF = ISHFT(WRDARR(IOFF+19),12) + WRDARR(IOFF+20)
        IF (I4BUFF .LT. 0) THEN

          I4BUFF = 65536 + I4BUFF
        ENDIF
        PRINT '("TIM1ST =",X,I8)', I4BUFF
C       Day of Last Major Frame
        PRINT '("DAYLST =",3X,I6)', WRDARR(IOFF+21)
C       Time (GMT) of Last Major Frame
        I4BUFF = ISHFT(WRDARR(IOFF+22),12) + WRDARR(IOFF+23)
        IF (I4BUFF .LT. 0) THEN
          I4BUFF = 65536 + I4BUFF
        ENDIF
        PRINT '("TIMLST =",X,I8)', I4BUFF
C       Number of Cse's on Orbit's Transmission
        PRINT '("NCSEOX =",3X,I6)', WRDARR(24)
C       Number of Cse's Read from Daily Tape
        PRINT '("NCSEOT =",3X,I6)', WRDARR(25)
C       Number of Cal Sequences in Orbit
        PRINT '("NCALSO =",3X,I6)', WRDARR(26)
      END DO
      PRINT '("------------------")'
C     EOR
      PRINT '("EOR    =",5X,O4.4)', WRDARR(IBLKSZ/2-1)
C     Checksum
      PRINT '("CHKSUM =",3X,I6)', WRDARR(IBLKSZ/2)

      RETURN
      END

C-----------------------------------------------------------------------
C     This Subroutine will print Day Header Record (5202)
C-----------------------------------------------------------------------
      SUBROUTINE PR5202(WRDARR, IBLKSZ, N)

      INTEGER*2          WRDARR(500)   ! Word Array
      CHARACTER          BUFF*4        ! Temporary data buffer

      PRINT '("******************")'

C     First two words are some kind of file marker
      PRINT '("SYNC   =",2(X,O4.4))', WRDARR(1:2)
C     Record length
      PRINT '("RECLEN =",3X,I6)', WRDARR(3)
C     Record number
      PRINT '("RECNUM =",3X,I6)', WRDARR(4)
C     Record identifier
      PRINT '("RECID  =",5X,O4.4)', WRDARR(5)
C     Day number
      PRINT '("DAYNUM =",3X,I6)', WRDARR(6)
C     Year
      PRINT '("YEAR   =",3X,I6)', WRDARR(7)
C     EOR
      PRINT '("EOR    =",5X,O4.4)', WRDARR(8)
C     Checksum
      PRINT '("CHKSUM =",3X,I6)', WRDARR(9)

      RETURN
      END

C-----------------------------------------------------------------------
C     This Subroutine will print the Day Header Record (5203)
C-----------------------------------------------------------------------
      SUBROUTINE PR5203(WRDARR, IBLKSZ, N)

      INTEGER*2          WRDARR(500)   ! Word Array
      CHARACTER          BUFF*4        ! Temporary data buffer
      INTEGER*4          I4BUFF        ! 4-byte integer buffer

      PRINT '("******************")'

C     First two words are some kind of file marker
      PRINT '("SYNC   =",2(X,O4.4))', WRDARR(1:2)
C     Record length
      PRINT '("RECLEN =",3X,I6)', WRDARR(3)
C     Record number
      PRINT '("RECNUM =",3X,I6)', WRDARR(4)
C     Record identifier
      PRINT '("RECID  =",5X,O4.4)', WRDARR(5)
C     Day number
      PRINT '("DAY    =",3X,I6)', WRDARR(6)
C     Year
      PRINT '("YEAR   =",3X,I6)', WRDARR(7)
C     Number of major frames in day
      I4BUFF = ISHFT(WRDARR(8),12) + WRDARR(9)
      PRINT '("NMFDAY =",3X,I6)', I4BUFF
C     Number of Cse's on Day's Transmission
      PRINT '("NCSEDX =",3X,I6)', WRDARR(10)
C     Number of Cse's Read from Daily Tape
      PRINT '("NCSEDT =",3X,I6)', WRDARR(11)
C     Number of Cal Sequences in Day
      PRINT '("NCALSD =",3X,I6)', WRDARR(12)
C     Number of Orbits in Day (max 16)
      PRINT '("NORBIT =",3X,I6)', WRDARR(13)
C     Calibration Data Used for the Processing of this Day's Obserations
      PRINT '("CALDAT =",2X,4(X,I6),19(/,10X,4(X,I6)))', WRDARR(14:93)
C     EOR
      PRINT '("EOR    =",5X,O4.4)', WRDARR(13)
C     Checksum
      PRINT '("CHKSUM =",3X,I6)', WRDARR(14)

      RETURN
      END

C-----------------------------------------------------------------------
C     This Subroutine will print the Orbit Header Record (5204)
C-----------------------------------------------------------------------
      SUBROUTINE PR5204(WRDARR, IBLKSZ, N)

      INTEGER*2          WRDARR(500)   ! Word Array
      CHARACTER          BUFF*4        ! Temporary data buffer
      INTEGER*4          I4BUFF        ! 4-byte integer buffer
      CHARACTER          HDRSS(2)

      PRINT '("******************")'

C     First two words are some kind of file marker
      PRINT '("SYNC   =",2(X,O4.4))', WRDARR(1:2)
C     Record length
      PRINT '("RECLEN =",3X,I6)', WRDARR(3)
C     Record number
      PRINT '("RECNUM =",3X,I6)', WRDARR(4)
C     Record identifier
      PRINT '("RECID  =",5X,O4.4)', WRDARR(5)
C     Orbit number
      I4BUFF = ISHFT(WRDARR(6),12) + WRDARR(7)
      IF (I4BUFF .LT. 0) THEN
        I4BUFF = 65536 + I4BUFF
      ENDIF
      PRINT '("ORBNUM =",X,I8)', I4BUFF
C     HDRSS 
      J = ISHFT(WRDARR(8),-6)
      HDRSS(1) = CHAR(IBCD(J))
      K = IAND(WRDARR(8),'003F'Z)
      HDRSS(2) = CHAR(IBCD(K))
      PRINT '("HDRSS  =",5X,2(X,A1))', HDRSS
C     Number of Major Frames in Orbit
      PRINT '("NMFORB =",3X,I6)', WRDARR(9)
C     Day of First Major Frame
      PRINT '("DAY1ST =",3X,I6)', WRDARR(10)
C     Time (GMT) of First Major Frame
      I4BUFF = ISHFT(WRDARR(11),12) + WRDARR(12)
      IF (I4BUFF .LT. 0) THEN
        I4BUFF = 65536 + I4BUFF
      ENDIF
      PRINT '("TIM1ST =",X,I8)', I4BUFF
C     Day of Last Major Frame
      PRINT '("DAYLST =",3X,I6)', WRDARR(13)
C     Time (GMT) of Last Major Frame
      I4BUFF = ISHFT(WRDARR(14),12) + WRDARR(15)
      IF (I4BUFF .LT. 0) THEN
        I4BUFF = 65536 + I4BUFF
      ENDIF
      PRINT '("TIMLST =",X,I8)', I4BUFF
C     Number of Cse's on Orbit's Transmission
      PRINT '("NCSEOX =",3X,I6)', WRDARR(16)
C     Number of Cse's Read from Daily Tape
      PRINT '("NCSEOT =",3X,I6)', WRDARR(17)
C     Number of Cal Sequences in Orbit
      PRINT '("NCALSO =",3X,I6)', WRDARR(18)
C     Number of Housekeeping Functions
      NHKFNC = WRDARR(19)
      PRINT '("NHKFNC =",3X,I6)', NHKFNC
C     Max. Housekeeping Functions
      PRINT '("MAXHK  =",2X,10(X,I6),5(/,10X,10(X,I6)))',
     +                                   WRDARR(0*NHKFNC+20:1*NHKFNC+19)
C     Min. Housekeeping Functions
      PRINT '("MINHK  =",2X,10(X,I6),5(/,10X,10(X,I6)))',
     +                                   WRDARR(1*NHKFNC+20:2*NHKFNC+19)
C     Mean Housekeeping Functions
      PRINT '("MEANHK =",2X,10(X,I6),5(/,10X,10(X,I6)))',
     +                                   WRDARR(2*NHKFNC+20:3*NHKFNC+19)
C     EOR

      PRINT '("EOR    =",5X,O4.4)', WRDARR(IBLKSZ/2-1)
C     Checksum
      PRINT '("CHKSUM =",3X,I6)', WRDARR(IBLKSZ/2)

      RETURN
      END

C-----------------------------------------------------------------------
C     This Subroutine will print the Data Record (5205)
C-----------------------------------------------------------------------
      SUBROUTINE PR5205(WRDARR, IBLKSZ, N)

      INTEGER*2          WRDARR(500)   ! Word Array
      CHARACTER          BUFF*4        ! Temporary data buffer
      INTEGER*4          I4BUFF        ! 4-byte integer buffer
      CHARACTER          HDRSS(2)

      PRINT '("******************")'

C     First two words are some kind of file marker
      PRINT '("SYNC   =",2(X,O4.4))', WRDARR(1:2)
C     Record length
      PRINT '("RECLEN =",3X,I6)', WRDARR(3)
C     Record number
      PRINT '("RECNUM =",3X,I6)', WRDARR(4)
C     Record identifier
      PRINT '("RECID  =",5X,O4.4)', WRDARR(5)
C     Number of major frames in this block (10 unless end of orbit)
      PRINT '("NMF    =",3X,I6)', WRDARR(6)
C     Length of entry for major frame (should be 186)
      PRINT '("MFLEN  =",3X,I6)', WRDARR(7)
C     Spare (set to zero)
      PRINT '("SPARE  =",3X,I6)', WRDARR(8)
      DO I=1,WRDARR(6)
        IOFF=(I-1)*WRDARR(7)
        PRINT '("------------------")'
C       Checksum word for major frame
        PRINT '("CKSUM  =",3X,I6)', WRDARR(IOFF+9)
C       Orbit number
        I4BUFF = ISHFT(WRDARR(IOFF+10),12) + WRDARR(IOFF+11)
        IF (I4BUFF .LT. 0) THEN
          I4BUFF = 65536 + I4BUFF
        ENDIF
        PRINT '("ORBNUM =",X,I8)', I4BUFF
C       Block number
        PRINT '("BLKNUM =",3X,I6)', WRDARR(IOFF+12)
C       HDRSS 
        J = ISHFT(WRDARR(IOFF+13),-6)
        HDRSS(1) = CHAR(IBCD(J))
        K = IAND(WRDARR(IOFF+13),'003F'Z)
        HDRSS(2) = CHAR(IBCD(K))
c       PRINT '("HDRSS  =",2(X,A1))', HDRSS
        PRINT '("FRMNUM =",3X,I6)', WRDARR(IOFF+13)

C       Day number
        PRINT '("DAYNUM =",3X,I6)', WRDARR(IOFF+14)

C       Time (GMT)
        I4BUFF = ISHFT(WRDARR(IOFF+15),12) + WRDARR(IOFF+16)
        IF (I4BUFF .LT. 0) THEN
          I4BUFF = 65536 + I4BUFF
        ENDIF
        PRINT '("TIME   =",X,I8)', I4BUFF
C       Latitude  (if over 90 subtract 2**12 to make negative, scaled by 8)
        I4BUFF = WRDARR(IOFF+17)
        IF (I4BUFF .GT. 90*8) THEN
          I4BUFF = I4BUFF - 2**12
        ENDIF
        PRINT '("LAT    =",X,F8.3)', REAL(I4BUFF)/8
C       Longitude (scaled by 8)
        PRINT '("LON    =",X,F8.3)', REAL(WRDARR(IOFF+18))/8
C       Altitude (scaled by 8 offset by 1000)
        PRINT '("ALT    =",X,F8.3)', REAL(WRDARR(IOFF+19))/8+1000
C       ESMR Maximum
        PRINT '("ESMRMX =",3X,I6)', WRDARR(IOFF+20)
C       ESMR Maximum
        PRINT '("ESMRMN =",3X,I6)', WRDARR(IOFF+21)
C       Flags
        PRINT '("FLAGS  =",5(X,B12.12))', WRDARR(IOFF+22:IOFF+26)
C       Calibrated Radiances
        PRINT '("CALRAD =",2X,10(X,I6),4(/,10X,10(X,I6)))',
     +                                         WRDARR(IOFF+27:IOFF+75)
C       16 sec Ramps
        PRINT '("RAMP16 =",2X,5(X,I6))', WRDARR(IOFF+76:IOFF+80)
C       4 sec Ramps
        PRINT '("RAMP4  =",2X,10(X,I6),4(/,10X,10(X,I6)))',
     +                                         WRDARR(IOFF+81:IOFF+124)
C       Digital A Housekeeping
        PRINT '("HKDIGA =",2X,5(X,I6))', WRDARR(IOFF+125:IOFF+129)
C       Analogue Housekeeping
        PRINT '("HKANLG =",2X,10(X,I6),3(/,10X,10(X,I6)))',
     +                                         WRDARR(IOFF+130:IOFF+168)
C       FOVC Ramp
        PRINT '("FOVCRP =",3X,I6)', WRDARR(IOFF+169)
C       Raw ESMR
        PRINT '("ESMR   =",2X,8(X,I6))', WRDARR(IOFF+170:IOFF+177)
C       Pitch
        PRINT '("PITCH  =",3X,I6)', WRDARR(IOFF+178)
C       Roll
        PRINT '("ROLL   =",3X,I6)', WRDARR(IOFF+179)
C       Yaw
        PRINT '("YAW    =",3X,I6)', WRDARR(IOFF+180)
C       A2, A3, A4 declouded
        PRINT '("ADECLD =",2X,3(X,I6))', WRDARR(IOFF+181:IOFF+183)
C       B1-B2, B2-B3, B3-B4 smoothed
        PRINT '("BSMTHD =",2X,3(X,I6))', WRDARR(IOFF+184:IOFF+186)
C       SST/GEOG
        PRINT '("SSTGEO =",3X,I6)', WRDARR(IOFF+187)
C       B1, B2, B3, B4, B1-B2, B2-B3, B3-B4 corrected radiances
        PRINT '("BCORAD =",2X,7(X,I6))', WRDARR(IOFF+188:IOFF+194)
      END DO
      PRINT '("------------------")'
C     EOR
      PRINT '("EOR    =",5X,O4.4)', WRDARR(IBLKSZ/2-1)

C     Checksum
      PRINT '("CHKSUM =",3X,I6)', WRDARR(IBLKSZ/2)

      RETURN
      END

C-----------------------------------------------------------------------
C     This Subroutine will print the End of Summary/Orbit/Day Records
C     (5202/5206/5207)
C-----------------------------------------------------------------------
      SUBROUTINE PR52EN(WRDARR, IBLKSZ, N)

      INTEGER*2          WRDARR(500)   ! Word Array
      CHARACTER          BUFF*4        ! Temporary data buffer

      PRINT '("******************")'

C     First two words are some kind of file marker
      PRINT '("SYNC   =",2(X,O4.4))', WRDARR(1:2)
C     Record length
      PRINT '("RECLEN =",3X,I6)', WRDARR(3)
C     Record number
      PRINT '("RECNUM =",3X,I6)', WRDARR(4)
C     Record identifier
      PRINT '("RECID  =",5X,O4.4)', WRDARR(5)
C     EOR
      PRINT '("EOR    =",5X,O4.4)', WRDARR(6)
C     Checksum
      PRINT '("CHKSUM =",3X,I6)', WRDARR(7)

      RETURN
      END

C-----------------------------------------------------------------------
C     This Function will return BCD to ASCII character index
C-----------------------------------------------------------------------
      FUNCTION IBCD(I)

      CHARACTER BCDTBL(64)
C                  _0  _1  _2  _3  _4  _5  _6  _7  
      DATA BCDTBL /' ','1','2','3','4','5','6','7', ! 0_
     +             '8','9','0','#','@','|','|','|', ! 1_
     +             '|','/','S','T','U','V','W','X', ! 2_
     +             'Y','Z','|',',','%','|','|','|', ! 3_
     +             '-','J','K','L','M','N','O','P', ! 4_
     +             'Q','R','|','$','*','|','|','|', ! 5_
     +             '&','A','B','C','D','E','F','G', ! 6_
     +             'H','I','|','.','_','|','|','|'/ ! 7_

      IF (BCDTBL(I+1) .EQ. '|') THEN
        IBCD = 0   ! Null
      ELSE IF (BCDTBL(I+1) .EQ. '_') THEN
        IBCD = 255 ! Square Lozenge Unavailabe in ASCII
      ELSE
        IBCD = ICHAR(BCDTBL(I+1))

      ENDIF

      RETURN
      END      END

Page | 30 


	1. Introduction
	1.1 Data Product Description
	1.1.1 The Selective Chopper Radiometer
	1.1.2 Nimbus-5 Overview
	1.2 Algorithm Background
	1.3 Data Disclaimer

	2. Data Organization
	2.1 File Naming Convention
	2.2 File Format and Structure
	2.3 Key Science Data Fields

	3. Data Contents
	3.1 Data Records
	3.2 Metadata

	4. Reading the Data
	5. Data Services
	5.1 GES DISC Search
	5.2 Documentation
	5.3 Direct Download

	6. More Information
	6.1 Contact Information
	6.2 References

	7. Appendices
	Acknowledgements
	Acronyms
	FORTRAN Code

